Pensieve Header: Some analytic BCH formulae in the blobs quotient - both symmetrization and de-symmetrization.

```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\w-Computations"];
<< "AlexanderBlobs-Program.m"
```

```
(b[r[1, 3], #] & /@ Table[r[j, 0], {j, 3}]) /. Diag[hs_, ars___] :> hs Diag[1, ars]
```

```
{0, 0, -Diag[1, h[0] t[3]] h[1] + Diag[1, h[0] t[1]] h[3]}
```

```
B[n_, expr_] := Module[
  {bra},
  Transpose[Table[
    bra = b[expr, r[i, 0]] /. Diag[hs_, ars___] :> hs Diag[1, ars];
    Coefficient[bra, r[#, 0]] & /@ Range[n],
    {i, n}
   ]]
  ];
EB3[expr_] := MatrixExp[B[3, expr]];
B[3, r[1, 3]] // MatrixForm
```

$$\begin{pmatrix} 0 & 0 & h[3] \\ 0 & 0 & 0 \\ 0 & 0 & -h[1] \end{pmatrix}$$

```
B3R[i_, j_] := MatrixExp[B[3, r[i, j]]];
B3R[1, 3] // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & \frac{e^{-h[1]}\left(-1+e^{h[1]}\right) h[3]}{h[1]} \\ 0 & 1 & 0 \\ 0 & 0 & e^{-h[1]} \end{pmatrix}$$

Symmetrization

```
(t1 = B3R[1, 3].B3R[2, 3]) // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & \frac{e^{-h[1]-h[2]}\left(-1+e^{h[1]}\right) h[3]}{h[1]} \\ 0 & 1 & \frac{e^{-h[2]}\left(-1+e^{h[2]}\right) h[3]}{h[2]} \\ 0 & 0 & e^{-h[1]-h[2]} \end{pmatrix}$$

```
(t2 = MatrixExp[B[3, F r[1, 3] + G r[2, 3]]]) // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & \frac{e^{-F h[1]-G h[2]}\left(-1+e^{F h[1]+G h[2]}\right) F h[3]}{F h[1]+G h[2]} \\ 0 & 1 & \frac{e^{-F h[1]-G h[2]}\left(-1+e^{F h[1]+G h[2]}\right) G h[3]}{F h[1]+G h[2]} \\ 0 & 0 & e^{-F h[1]-G h[2]} \end{pmatrix}$$

```
Solve[
 Thread[Equal[Flatten[t1], Flatten[t2]]] /. h[3] → 1,
 {F, G}
]
```

Solve::nsmet : This system cannot be solved with the methods available to Solve. ≫

$$\text{Solve}\left[\left\{\text{True}, \text{True}, \frac{e^{-h[1]-h[2]}\left(-1+e^{h[1]}\right)}{h[1]} == \frac{e^{-F h[1]-G h[2]}\left(-1+e^{F h[1]+G h[2]}\right) F}{F h[1]+G h[2]}, \text{True}, \text{True}, \frac{e^{-h[2]}\left(-1+e^{h[2]}\right)}{h[2]} == \frac{e^{-F h[1]-G h[2]}\left(-1+e^{F h[1]+G h[2]}\right) G}{F h[1]+G h[2]}, \text{True}, \text{True}, e^{-h[1]-h[2]} == e^{-F h[1]-G h[2]}\right\}, \{F, G\}\right]$$

```
Solve[{a x + b == 0, True}, x]
```

$$\left\{\left\{x \to -\frac{b}{h[4]\, t[1]+h[5]\, t[2]+h[6]\, t[3]}\right\}\right\}$$

```
eq1 = DeleteCases[Thread[Equal[Flatten[t1], Flatten[t2]]], True]
```

$$\left\{\frac{e^{-h[1]-h[2]}\left(-1+e^{h[1]}\right) h[3]}{h[1]} == \frac{e^{-F h[1]-G h[2]}\left(-1+e^{F h[1]+G h[2]}\right) F h[3]}{F h[1]+G h[2]}, \frac{e^{-h[2]}\left(-1+e^{h[2]}\right) h[3]}{h[2]} == \frac{e^{-F h[1]-G h[2]}\left(-1+e^{F h[1]+G h[2]}\right) G h[3]}{F h[1]+G h[2]}, e^{-h[1]-h[2]} == e^{-F h[1]-G h[2]}\right\}$$

```
Solve[-h[1] - h[2] == -F h[1] - G h[2], G]
```

$$\left\{\left\{G \to \frac{h[1]-F h[1]+h[2]}{h[2]}\right\}\right\}$$

$$\text{eq2} = \text{DeleteCases}\left[\text{Simplify}\left[\text{eq1} \;/.\; \left\{G \to \frac{h[1]-F h[1]+h[2]}{h[2]}, h[3] \to 1\right\}\right], \text{True}\right]$$

$$\left\{\frac{e^{-h[1]-h[2]}\left(-1+e^{h[1]}\right)}{h[1]} == \frac{F-e^{-h[1]-h[2]} F}{h[1]+h[2]}, \frac{e^{-h[1]-h[2]}\left(\left(1-e^{h[1]}-F+e^{h[1]+h[2]} F\right) h[1]-\left(-1+e^{h[1]}\right) h[2]\right)}{h[2]\,(h[1]+h[2])} == 0\right\}$$

$$\text{Solve}\left[\frac{e^{-h[1]-h[2]}\left(-1+e^{h[1]}\right)}{h[1]} == \frac{F-e^{-h[1]-h[2]} F}{h[1]+h[2]}, F\right]$$

$$\left\{\left\{F \to \frac{\left(-1+e^{h[1]}\right)(h[1]+h[2])}{\left(-1+e^{h[1]+h[2]}\right) h[1]}\right\}\right\}$$

$$\text{Simplify}\left[\frac{e^{-h[1]-h[2]}\left(\left(1-e^{h[1]}-F+e^{h[1]+h[2]} F\right) h[1]-\left(-1+e^{h[1]}\right) h[2]\right)}{h[2]\,(h[1]+h[2])} \;/.\; F \to \frac{\left(-1+e^{h[1]}\right)(h[1]+h[2])}{\left(-1+e^{h[1]+h[2]}\right) h[1]}\right]$$

0

```
Simplify[((-1 + ⅇ^h[1]) (h[1] + h[2]))/((-1 + ⅇ^(h[1]+h[2])) h[1]) /. {h[1] → x, h[2] → y}]
```

$$\frac{(-1+e^{x})(x+y)}{(-1+e^{x+y})\,x}$$

```
Simplify[(h[1] - F h[1] + h[2])/h[2] /. {F → ((-1 + ⅇ^x) (x + y))/((-1 + ⅇ^(x+y)) x), h[1] → x, h[2] → y}]
```

$$\frac{e^{x}(-1+e^{y})(x+y)}{(-1+e^{x+y})\,y}$$

```
Simplify[eq1 /. {F → ((-1 + ⅇ^x) (x + y))/((-1 + ⅇ^(x+y)) x), G → (ⅇ^x (-1 + ⅇ^y) (x + y))/((-1 + ⅇ^(x+y)) y), h[1] → x, h[2] → y}]
```

{True, True, True}

```
FullSimplify[{((-1 + ⅇ^x) (x + y))/((-1 + ⅇ^(x+y)) x), (ⅇ^x (-1 + ⅇ^y) (x + y))/((-1 + ⅇ^(x+y)) y)} /. {x → -y, y → -x}]
```

$$\left\{\frac{e^{x}(-1+e^{y})(x+y)}{(-1+e^{x+y})\,y},\ \frac{(-1+e^{x})(x+y)}{(-1+e^{x+y})\,x}\right\}$$

De-Symmetrization

```
(s1 = EB3[α r[1, 3] + β r[2, 3]]) // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & \frac{e^{-\alpha h[1]-\beta h[2]}\left(-1+e^{\alpha h[1]+\beta h[2]}\right)\alpha h[3]}{\alpha h[1]+\beta h[2]} \\ 0 & 1 & \frac{e^{-\alpha h[1]-\beta h[2]}\left(-1+e^{\alpha h[1]+\beta h[2]}\right)\beta h[3]}{\alpha h[1]+\beta h[2]} \\ 0 & 0 & e^{-\alpha h[1]-\beta h[2]} \end{pmatrix}$$

```
(s2 = EB3[ξ r[1, 3]].EB3[η r[2, 3]]) // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & \frac{e^{-\xi h[1]-\eta h[2]}\left(-1+e^{\xi h[1]}\right) h[3]}{h[1]} \\ 0 & 1 & \frac{e^{-\eta h[2]}\left(-1+e^{\eta h[2]}\right) h[3]}{h[2]} \\ 0 & 0 & e^{-\xi h[1]-\eta h[2]} \end{pmatrix}$$

```
sol2 = Solve[
   Thread[Equal[Flatten[s1], Flatten[s2]]] /. {h[1] → x, h[2] → y, h[3] → 1},
   {ξ, η}
  ] // FullSimplify
```

Solve::ifun : Inverse functions are being used by Solve, so
some solutions may not be found; use Reduce for complete solution information. ≫

$$\left\{\left\{\xi \to \frac{-\mathrm{Log}\left[e^{-x\alpha-y\beta}\right]+\mathrm{Log}\left[\frac{x\alpha+e^{-x\alpha-y\beta}y\beta}{x\alpha+y\beta}\right]}{x},\ \eta \to -\frac{\mathrm{Log}\left[\frac{x\alpha+e^{-x\alpha-y\beta}y\beta}{x\alpha+y\beta}\right]}{y}\right\}\right\}$$

```
(Exp[{ξ, η}] /. sol2) // FullSimplify
```

$$\left\{\left\{e^{\frac{-\mathrm{Log}\left[e^{-x\alpha-y\beta}\right]+\mathrm{Log}\left[\frac{x\alpha+e^{-x\alpha-y\beta}y\beta}{x\alpha+y\beta}\right]}{x}},\ \left(\frac{x\alpha+e^{-x\alpha-y\beta}y\beta}{x\alpha+y\beta}\right)^{-1/y}\right\}\right\}$$

```
({ξ, η} /. sol2 /. {α → 1, β → 1}) // FullSimplify
```

$$\left\{\left\{\frac{-\text{Log}\left[e^{-x-y}\right]+\text{Log}\left[\frac{x+e^{-x-y}\,y}{x+y}\right]}{x},\ -\frac{\text{Log}\left[\frac{x+e^{-x-y}\,y}{x+y}\right]}{y}\right\}\right\}$$